


Knud Møller

Mikro-Logo



L. R. DATA 07-154711

FORSLAG TIL LØSNINGER

Dette hæfte indeholder:

Forslag til løsninger på øvelserne i "MIKRO-LOGO".

Forslag til ekstraopgaver.

Den nyeste version af mikrologo-sproget indeholder væsentlige forbedringer af sproget i forhold til tidligere versioner.

Af forbedringer kan nævnes:

Tegnehastigheden er øget både med og uden pil.

Rekursive procedurer kan udføres flere gange uden at lagerkapaciteten overskrides.

Når plist, pdisk, phjælp og kopi anvendes første gang spørges: Hvilket skriver nummer: _

Middelfart den 16. oktober 1985

Knud Møller

Forfatter:
Knud Møller

Udgiver:
L.R.Data, Ikast

MIKRO-LOGO.

Kapitel 1: Simple ordrer.

Øvelse 5.

Trekanten udføres lettest ved at anvende hjem-sætningen, f.eks. således:

start;frem 70;venstre 50;frem 100;hjem

Øvelse 9.

Selvom procedure-begrebet ikke er kendt, bør problemet løses ved at dele opgaven op i en række underopgaver:

tegn yderste kasse (rombe)
flyt pilen ind i kassen
tegn inderste kasse
slet inderste kasse

Eleven kan allerede nu gøre sig nogle overvejelser angående vinklerne i en rombe.

Løsningsforslag:

start
fr 140;ve 60;fr 140;ve 120;fr 140;ve 60;fr 140;ve 120
løft
fr 20;ve 60;fr 20;hø 60;tegn
fr 100;ve 60;fr 100;ve 120;fr 100;ve 60;fr 100;ve 120
slet
fr 100;ve 60;fr 100;ve 120;fr 100;ve 60;fr 100;ve 120

Øvelse 9 vil eventuelt kunne tages op efter indførelse af procedurer.

Øvelse 10.

Kasserne:

gentag 4;fr 40;ve 90;hertil
gentag 4;fr 80;ve 90;hertil
gentag 4;fr 120;ve 90;hertil

Trekanterne:

gentag 3;gentag 3;fr 80;ve 120;hertil;ve 120;hertil

N'et:

start;løft;tb 70;ve 90;tegn;fr 150;hjem;ve 90;fr 150

Tre-tallet:

gentag 2;gentag 18;fr 5;hø 10;hertil;hø 180;hertil

Kvadratet:

start

ve 45;gentag 4;fr 50;ve 90;hertil

Ordren drejtil og ordren status kan eventuelt indføres på nuværende tidspunkt, hvis der opstår et behov.

Kapitel 2: Procedurer.

Øvelse 23.

begynd sekskant gentag 6 fr 30 ve 60 hertil slut

begynd ottekant gentag 8 fr 30 ve 45 hertil slut

begynd attenkant gentag 18 fr 20 ve 360/18 hertil slut

Øvelse 25.

6 gange.

Øvelse 25.

højre 20 udskiftes med hø 60

Øvelse 29.

Løsningsforslag:

begynd opg29 gentag 6 sekskant fr 30 hø 60 hertil slut
--

Øvelse 30.

Løsningsforslag:

```

begynd opg30
  gentag 6
    fr 10
    ve 60
    fr 10
    ve 60
    fr 10
    hø 60
  hertil
slut

```

Programmet i øvelse 30 kan anvendes til et mønster. Her er vist, hvorledes 2 figurer lægges sammen:

```

begynd mønster30
  start
  opg30
  fr 10
  hø 180
  opg30
  slut

```

Nogle eksempler på anvendelse af proceduren, som tegner ottekanter:

```

begynd ottere1
  start
  gentag 4
    ottekant
    hø 90
  hertil
slut

```

```

begynd ottere2
  start
  gentag 8
    ottekant
    hø 45
  hertil
slut

```

```

begynd ottere3
  start
  gentag 8
    ottekant
    hø 45
    fr 30
  hertil
slut

```

```

begynd ottere4
  start
  gentag 8
    ottekant
    fr 30
    hø 45
  hertil
slut

```

```

begynd ottere5
  start
  gentag 8
    ottekant
    fr 30
    hø 90
    fr 30
    ve 45
  hertil
slut

```

Kapitel 3: Variable.Øvelse 32.

Løsningsforslag:

```
begynd opg32
start
gentag 10
ve 30
tre 100
hø 30
løft
fr 20
tegn
hertil
slut
```

Forslag til ekstraoppgave:

Procedure, hvor afstanden mellem trekkanterne indgår en variabel.

Løsningsforslag:

```
begynd opg321 afstand
start
gentag 10
ve 30
tre 100
hø 30
løft
fr afstand
tegn
hertil
slut
```

Forslag til ekstraoppgave:

Procedure, hvor både afstanden mellem trekkanterne og antallet af trekkanter indgår.

Løsningsforslag:

```
begynd opg322 afstand antal
start
gentag antal
ve 30
tre 100
hø 30
løft
fr afstand
tegn
hertil
slut
```

Ovenstående ekstraoppgave kan eventuelt udvides til at omfatte trekantens størrelse.

Forslag til ekstraopgave:

Procedure, hvor trekkanterne drejes i forhold til den foregående.

Løsningsforslag:

```
begynd opg323 afstand antal vinkel
start
gentag antal
  ve 30
  tre 100
  hø 30
  løft
  fr afstand
  tegn
  venstre vinkel
  hertil
slut
```

Øvelse 36.

Løsningsforslag:

```
begynd rekt længde bredde
gentag 2
  fr længde
  ve 90
  fr bredde
  ve 90
  hertil
slut
```

Øvelse 37.

Løsningsforslag:

```
begynd opg37
sætning tast længde på rektanglet
hentall længde
sætning tast bredde på rektanglet
hentall bredde
gentag 2
  fr længde
  ve 90
  fr bredde
  ve 90
  hertil
slut
```

Øvelse 39.

Løsningsforslag:

```
begynd opg39 hvilken side
gentag hvilken
fr side
ve 360/hvilken
hertil
slut
```

Øvelse 40.

```
begynd opg40
sætning hvilken -kant ønskes tegnet ?
henttal hvilken
sætning med hvilken sidelængde ?
henttal side
gentag hvilken
fr side
ve 360/hvilken
hertil
slut
```


Øvelse 41.

Her er vist tre løsningsforslag:

```
begynd opg41nr1 u v side
fr side
ve v
fr side
ve u
fr side
ve v
fr side
ve u
slut
```

```
begynd opg41nr2 u v side
gentag 2
fr side
ve v
fr side
ve u
hertil
slut
```

```
begynd opg41nr3 v side
gentag 2
fr side
ve v
fr side
ve 180-v
hertil
slut
```


Øvelse 43.

```

begynd opg43 vinkel længde
start
opg41nr3 vinkel længde
fr længde
ve vinkel/2
fr længde/2
ve vinkel/2
fr længde
hø vinkel/2
tb længde/2
hø vinkel/2
tb længde
ve vinkel/2
fr længde/2
hø vinkel/2
fr længde
slut

```

Kapitel 4: Tildelingssætninger.Øvelse 45.

Hvis trekkanterne ønskes slettet i samme rækkefølge, som de er tegnet, kan denne procedure anvendes:

```

begynd opg45
nytre
slet
nytre
tegn
slut

```

Hvis de største trekkanter ønskes slettet først, kan disse procedurer anvendes. Proceduren nytrel udnytter, at side-værdien er kendt ved procedure-kaldet.

```

begynd opg451
nytre
slet
nytrel
tegn
slut

```

```

begynd nytrel
gentag 15
tre side
side :=side-10
hertil
slut

```

Øvelse 46.

Nogle eksempler:

```

begynd nycirkel1
side :=1
gentag 30
  cirkel side
  side :=side+1
  ve 5
hertil
slut

```

```

begynd cirkel side
gentag 36
  fr side
  ve 10
  hertil
slut

```

```

begynd nycirkel
side :=1
gentag 30
  cirkel side
  side :=side+1
  hertil
slut

```

Øvelse 47.

To eksempler på anvendelse af tildelingsætninger ved tegning af rhomber:

```

begynd rombe1
side :=160
vinkel :=50
gentag 32
  gentag 2
    fr side
    ve vinkel
    fr side
    ve 180-vinkel
  hertil
  side :=side-10
hertil
slut

```

```

begynd rombe
side :=140
vinkel :=10
gentag 36
  gentag 2
    fr side
    ve vinkel
    fr side
    ve 180-vinkel
  hertil
  vinkel :=vinkel+10
hertil
slut

```

```

begynd rombe2
start
side :=160
vinkel :=50
gentag 64
  gentag 2
    fr side
    ve vinkel
    fr side
    ve 180-vinkel
  hertil
  side :=side-5
  vinkel :=vinkel+10
hertil
slut

```

Et eksempel ved tegning af rektangler:

```

begynd nyrekt
start
l :=200
b :=5
gentag 20
  rekt l b
  l :=l-10
  b :=b+5
hertil
slut

```

```

begynd rekt l b
gentag 2
  fr l
  ve 90
  fr b
  ve 90
hertil
slut

```

Op-gave 42 fra kapitel 3 kan eventuelt tages op, idet mønster-tegningen naturligvis lettes ved brug af variable.

Kapitel 6: Regning med logo.Løsningsforslag til øvelserne 54 - 61.

```

begynd femtabel
x :=5
gentag 10
  skriv x
  x :=x+5
hertil
slut

begynd baglæns
x :=100
gentag 100
  skriv x
  x :=x-1
hertil
slut

begynd kvadrat
x :=1
gentag 10
  skriv x**2
  x :=x+1
hertil
slut

begynd addition
sætning indtast første tal
henttal x
sætning indtast andet tal
henttal y
sum :=x+y
sætning summen af
skriv x
sætning og
skriv y
sætning er:
skriv sum
slut

```

```

begynd potens
sætning indtast roden
henttal rod
sætning indtast eksponenten
henttal eks
pot :=rod**eks
skriv rod
sætning opløftet i
skriv eks
sætning 'de potens er
skriv pot
slut

```

```

begynd middeltal
sætning tast første tal
henttal x
sætning tast andet tal
henttal y
sætning tast tredje tal
henttal z
sum :=x+y+z
middel :=sum/3
sætning middeltallet er
skriv middel
slut

```

```

begynd løn
sætning tast timeløn
henttal tløn
sætning tast timetal
henttal timer
sætning lønnen er
skriv tløn*timer
slut

```

```

begynd rekt
sætning indtast rektanglets
henttal længde
hvis ((længde>32766)+(længde<=0))<>0 -->Simulering af den logiske
operator OR (eller).
stop SAND = 1 FALSK = 0
sluthvis Se manualen side 10.
breddemax :=32766/længde
sætning bredden må ikke overstige
skriv breddemax
sætning tast et bogstav for at fortsætte
henttast q
sætning indtast rektanglets
henttal bredde
hvis ((bredde>breddemax)+(bredde<0))<>0
stop
ellers
sætning rektanglets areal er ....
skriv længde*brede
sætning arealenheder
sluthvis
slut

```


Kapitel 6: Anvendelse af sætx, sæty og sætxy. Grafer.Øvelse 66.

Linierne kan f.eks. ændres til:

```
x:=tlf(450)-225 og y:=tlf(400)-200
```

Øvelse 67.

sæt x flytter pilen vandret til den angivne x -koordinat.
sæt y flytter pilen lodret til den angivne y -koordinat.

Øvelse 68.

Vigtigt: Hvis potensopløftningen (**) anvendes, skal funktionsudtrykket skrives således:

```
y:=(x**2)/10
=====
```

Øvelse 69.

Løsningsforslag:

```

begynd opg691
x :=-101
gentag 100
  x :=x+2
  y :=(x**2)/10
  sætxy x y
hertil
slut

```

```

begynd opg692
løft
x :=-100
y :=(x**2)/10
sætxy x y
tegn
gentag 100
  x :=x+2
  y :=(x**2)/10
  sætxy x y
hertil
slut

```

Øvelse 70.

Forslag til akse-procedure:

```

begynd akser
start
løft
sætxy -250 0
tegn
sæt $x$  250
løft
sætxy 0 -200
tegn
sæt $y$  200
slut

```


Hvis der skal afsættes enheder på akserne kan disse procedurer anvendes:

Hovedprocedure:

```

begynd aksermind
  akser
  xakseind
  yakseind
slut
  
```

Underprocedurerne xakseind og yakseind ser således ud:

```

begynd xakseind
  x :=-200
  gentag 40
    xmærker x
    x :=x+10
  hertil
slut
  
```

```

begynd yakseind
  y :=-200
  gentag 40
    ymærker y
    y :=y+10
  hertil
slut
  
```

xakseind og yakseind udnytter henholdsvis xmærker og ymærker:

```

begynd xmærker x
  løft
  sætxy x -3
  tegn
  sæt y 3
slut
  
```

```

begynd ymærker y
  løft
  sætxy -3 y
  tegn
  sæt x 3
slut
  
```

Øvelse 73.

Løsningsforslag:

```

begynd linie1
  x :=-100
  y :=3*x-10
  løft
  sætxy x y
  tegn
  gentag 100
    x :=x+2
    y :=3*x-10
    sætxy x y
  hertil
slut
  
```

Øvelse 74.

Løsningsforslag:

```

begynd linie2 a b
  x :=-100
  y :=a*x+b
  løft
  sætxy x y
  tegn
  gentag 100
    x :=x+2
    y :=a*x+b
    sætxy x y
  hertil
slut
  
```

Da mikrologoen ikke arbejder med decimaltal, er man nødt til at gå en lille omvej for at kunne tegne linier af formen: $y = 0.25 * x + 10$

Problemet kan løses ved dette program:

```

begynd linie3
  sætning indtast tæller i a-værdien
  henttal t
  sætning indtast nævner i a-værdien
  henttal n
  sætning indtast b-værdien
  henttal b
  x :=-100
  y :=(t*x)/n+b
  løft
  sætxy x y
  tegn
  gentag 100
    x :=x+2
    y :=(t*x)/n+b
    sætxy x y
  hertil
slut

```

Øvelse 78.

Løsningsforslag:

```

begynd hyp78
  aksermind
  x :=-151
  y :=1000/x
  løft
  sætxy x y
  tegn
  gentag 150
    x :=x+2
    y :=1000/x
    sætxy x y
  hertil
slut

```

Øvelse 79.

Løsningsforslag:

```

begynd hyp79
  tal :=4000
  aksermind
  gentag 20
    x :=-201
    y :=tal/x
    løft
    sætxy x y
    tegn
  gentag 200
    x :=x+2
    y :=tal/x
    sætxy x y
  hertil
  tal :=tal-400
  hertil
slut

```

Kapitel 8: Rekursive procedurer.

Vigtigt: Den nyeste version af mikrologo arbejder langt bedre end de tidligere version, bl.a. kommer meddelelsen om ikke mere lager på et langt senere tidspunkt. Dette muliggør nogle bedre rekursive procedurer.

Til sidst i dette afsnit er der et forslag til flere rekursive procedurer.

Øvelse 82.Løsningsforslag:

```

begynd ny82 side vinkel
  fr side
  ve vinkel
  ny82 side vinkel
slut

```

Øvelse 84.

```

begynd polydrej side v d
  hvis side>100
    stop
  ellers
    frem side
    venstre v
    polydrej side+d v d
  sluthvis
slut

```

Øvelse 85.

```

begynd opg side
  hvis side>160
    stop
  ellers
    gentag 3
      frem side
      ve 120
    hertil
    ve 5
  opg side+4
  sluthvis
slut

```

Øvelse 86.

```

begynd opg86
  start
  opg 0
  slet
  slette 164
slut

```

```

begynd slette side
  hvis side<0
    stop
  ellers
    gentag 3
      frem side
      ve 120
    hertil
    hø 5
    slette side-4
  sluthvis
slut

```


Forslag til ekstraopgave:

Udvid procedurerne til mere generelle procedurer.

Øvelse 87.

Se løsningsforslag i manualen.

Forslag til ekstraopgave:

Proceduren polydrej arbejder med en ændring af værdien x , altså ændring af en længde. Nedenstående procedure bevirker en ændring af vinklen ved hvert nyt procedure-kald.

```
begynd polyvinkel x v d
  frem x
  venstre v
  polyvinkel x v+d d
  slut
```

Afprøv proceduren. Der kan virkelig laves særdeles spændende figurer.

Forslag til ekstraopgave:

polydrej og polyvinkel kan 'samkøres', således at både længden og vinklen ændres ved procedure-kaldet.

```
begynd polyvinkogside x v d t
  fr x
  ve v
  polyvinkogside x+d v+t d t
  slut
```

Forslag til ekstraopgave:

Overvej procedurer i stil med polydrej og polyvinkel. Indsæt andre (og flere) sætninger i stedet for frem x og venstre v.

Kapitel 9: Ekstraopgaver.Øvelse 89.

Nedenstående procedurer klarer spejlingen:

```
begynd spejlakse
løft
sætxy 0 -200
tegn
sætxy 200
hjem
slut
```

```
begynd hvilkenfig
sætning hvilken -kant ønskes spejlet ?
henttal kant
sætning angiv sidelængde i figuren
henttal side
sætning placering af figuren
henttal x
henttal y
sætning angiv drejning af figuren i forhold til vandret
henttal drejv
tegnfig kant side x y drejv
slut
```

```
begynd tegnfig kant side x y drejv
løft
sætxy x y
drejtil drejv
tegn
gentag kant
fr side
ve 360/kant
hertil
slut
```

```
begynd udførspejl
løft
sætxy -x y
drejtil 180-drejv
tegn
gentag kant
fr side
hø 360/kant
hertil
slut
```

```
begynd spejling
spejlakse
hvilkenfig
sætning spejlingen udføres når en tast berøres
henttast q
udførspejl
slut
```

Øvelse 90.

Procedurer som udfører en drejning:

```

begynd udfør1
løft
sætxy x0 y0
drejtil drejvinkel
frem x-x0
ve 90
frem y-y0
drejtil drejvinkel+figurdrej
tegn
tegnfig kant side
sætning ønskes forbindelseslinie tegnet ? 1 = ja
henttal svar
hvis svar=1
  sætxy x0 y0
  sætxy x y
  sætxy x0 y0
sluthvis
slut

```

```

begynd drejning1
drejpkt1
hvilkenfigdrej
retnogstør
udfør1
slut

```

```

begynd drejpkt1
sætning indtast koordinater for drejningspunktet
henttal x0
henttal y0
løft
sætxy x0-2 y0-2
drejtil 0
tegn
gentag 4
  fr 4
  ve 90
hentil
slut

```

```

begynd tegnfig kant side
gentag kant
  fr side
  ve 360/kant
hentil
slut

```

```

begynd hvilkenfigdrej
sætning hvilken -kant ønskes drejet ?
henttal kant
sætning angiv sidelængde i figuren
henttal side
sætning placering af figuren
henttal x
henttal y
sætning drejning af fig. i forhold til -----> = 0 grader
henttal figurdrej
løft
sætxy x y
drejtil figurdrej
tegn
tegnfig kant side
slut

```

```

begynd retnogstør
sætning drejning til venstre: pos. tal til højre: neg. tal
henttast q
sætning tast drejningsvinklen
henttal drejvinkel
slut

```

Øvelse 91.

Procedurer, som udfører en parallelforskydning:

```

begynd retnogafst
  sætning retning for parallelforskydningen i forhold til vandret
  henttal retning
  sætning tast længde på parallelforskydningen
  henttal længde
slut

```

```

begynd tegnfig kant side
  gentag kant
    fr side
    ve 360/kant
  her til
slut

```

```

begynd parallelfor
  hvilkenfig
  tegnfig kant side
  retnogafst
  sætning parallelforskydningen udføres ved at taste et bogstav
  henttast q
  udfør
slut

```

```

begynd udfør
  drejtil retning
  løft
  frem længde
  drejtil figurdrej
  hvor x1 y1 ret
  tegn
  tegnfig kant side
  løft
  sætxy x y
  tegn
  sætxy x1 y1
slut

```

```

begynd hvilkenfig
  sætning hvilken -kant ønskes parallelforskydning ?
  henttal kant
  sætning angiv sidelængde i figuren
  henttal side
  sætning placering af figuren
  henttal x
  henttal y
  sætning drejning af fig. i forhold til ----> = 0 grader
  henttal figurdrej
  løft
  sætxy x y
  drejtil figurdrej
  tegn
  tegnfig kant side
slut

```


Øvelse 92.Løsningsforslag:

```


begynd cirkel r
løft
fr r
ve 90
side :=r*2*22/7/36+1
tb side/2
tegn
gentag 36
  frem side
  ve 10
hertil
løft
fr side/2
hø 90
tb r
tegn
slut

```

Hvis cirkelns centrum ønskes med ved procedurekaldet foretages disse ændringer:

1. linie ændres til: `begynd cirkel2 r xstart ystart`
 Disse linier indføjes:
 Som linie 3: `sætxy xstart ystart`
 Som linie 4: `drejtil 0`

NB ! På grund af den manglende mulighed for at kunne arbejde med decimaltal, kan cirklerne blive en smule unøjagtige. Dette viser sig ved en række koncentriske cirkler.



Øvelse 93.

Knæk-kurverne kan laves ved hjælp af en række simple procedurer, som udnytter 'de foregående'. Desuden er vist proc5, der tegner knækkurverne endnu mere findelt

```

begynd proc1 x
  fr x
  slut

begynd proc2 x
  proc1 x/3
  ve 60
  proc1 x/3
  hø 120
  proc1 x/3
  ve 60
  proc1 x/3
  slut

begynd proc3 x
  proc2 x/3
  ve 60
  proc2 x/3
  hø 120
  proc2 x/3
  ve 60
  proc2 x/3
  slut

```

```

begynd proc4 x
  proc3 x/3
  ve 60
  proc3 x/3
  hø 120
  proc3 x/3
  ve 60
  proc3 x/3
  slut

begynd proc5 x
  proc4 x/3
  ve 60
  proc4 x/3
  hø 120
  proc4 x/3
  ve 60
  proc4 x/3
  slut

```

Hvis knæk-kurverne ønskes placeret et andet sted på skærmen, kan der anvendes start-procedurer. Her er vist nogle procedurer, som anvendes i det følgende.

```

begynd klar
  start
  løft
  sætxy -200 -50
  tegn
  slut

```

```

begynd klar2
  start
  løft
  sætxy -200 -100
  tegn
  slut

```

```

begynd klar3
  start
  løft
  sætxy 50 50
  tegn
  slut

```

Disse startprocedurer kunne naturligvis udføres som en procedure:

```

begynd startklar x y
  start
  løft
  sætxy x y
  tegn
  slut



```

Hvis knæk-kurverne anvendes i andre procedurer, kan der laves spændende mønstre.

Her er nogle eksempler:

```
begynd forsøg  
klar  
gentag 3  
  proc4 243  
  ve 120  
hertil  
slut
```


```
begynd forsøg2  
klar2  
gentag 4  
  proc4 243  
  ve 90  
hertil  
slut
```



Forslag til ekstraopgaver:

Lav knæk-kurver med andre 'grundlinier' og anvend disse procedurer i procedurer i stil med 'forsøg' og 'forsøg2'.

Her er et eksempel:



```


begynd proc22 x
  proc1 x/3
  ve 90
  proc1 x/3
  hø 120
  proc1 x/3
  ve 90
  proc1 x/3
slut
  
```

```

begynd proc32 x
  proc22 x/3
  ve 60
  proc22 x/3
  hø 120
  proc22 x/3
  ve 60
  proc22 x/3
slut
  
```

```

begynd proc42 x
  proc32 x/3
  ve 60
  proc32 x/3
  hø 120
  proc32 x/3
  ve 60
  proc32 x/3
slut
  
```




```

begynd forsøgs
  v :=0
  gentag 6
  løft
  hjem
  drejt:1 v
  tegn
  proc42 243
  v :=v+60
  hertil
slut
  
```

```

begynd forsøg7
  v :=0
  gentag 6
  løft
  hjem
  drejt:1 v
  frem 20
  tegn
  proc42 243
  v :=v+60
  hertil
slut
  
```



```

begynd forsøg5
  klar3
  gentag 3
  proc42 243
  hertil
slut
  
```

```

begynd klar3
  start
  løft
  sætxy 50 50
  tegn
slut
  
```


Geometric Figures

Draw the following figures on a grid. Each figure is composed of line segments of equal length.



Figure 1

Figure 2

Draw the following figure on a grid. Each figure is composed of line segments of equal length.

Figure 3

Draw the following figure on a grid. Each figure is composed of line segments of equal length.

Figure 4

Draw the following figure on a grid. Each figure is composed of line segments of equal length.




Figure 5

Draw the following figure on a grid. Each figure is composed of line segments of equal length.

Figure 6

Draw the following figure on a grid. Each figure is composed of line segments of equal length.

Figure 7

Draw the following figure on a grid. Each figure is composed of line segments of equal length.

Figure 8

Draw the following figure on a grid. Each figure is composed of line segments of equal length.

Figure 9

Draw the following figure on a grid. Each figure is composed of line segments of equal length.

Figure 10

Draw the following figure on a grid. Each figure is composed of line segments of equal length.

